

Лабораторная работа по Excel

(файл .xls на странице www.matburo.ru/sub_appear.php?p=l_excel)

Тема: Работа с массивами

ЗАДАНИЕ

1. По заданным координатам точек A, B, C, D найти координаты векторов $a=AB$ и $b=CD$.
2. Вычислить скалярное произведение найденных векторов.
3. Найти следующие произведения векторов на заданную матрицу M : $a*M$ и $M*b$.
4. Вычислить определители матриц M и S .
5. Найти обратные матрицы S^{-1} и M^{-1} .
6. Вычислить произведение матрицы S на обратную к ней S^{-1} .
7. Найти решение системы линейных уравнений $Sx=b$ и $Mx=a$.
8. Выполнить проверку для найденных решений.
9. Сохранить документ.

Данные из таблицы 3.1:

Матрица системы

$$A = \begin{pmatrix} -3 & 1 & 3 \\ 0 & -2 & -1 \\ 2 & -1 & 3 \end{pmatrix}$$

Вектор правой части

$$f = \begin{pmatrix} 10 \\ -4 \\ 3 \end{pmatrix}$$

РЕШЕНИЕ.

В приведенных данных нет координат точек, нужных для решения задачи. Возьмем точки из примера.

$$A=(2, -1, 0); B=(-1, 4, -6); C=(1, -3, -5); D=(-2, -1, 0);$$

В условии приведена матрица A. Будем ее использовать в качестве матрицы M.

$$M = \begin{pmatrix} -3 & 1 & 3 \\ 0 & -2 & -1 \\ 2 & -1 & 3 \end{pmatrix}$$

Матрицы S в условии нет. Для выполнения задания будем использовать матрицу из примера.

$$S = \begin{pmatrix} 1 & -1 & 8 \\ 3 & 3 & -5 \\ -2 & -7 & 11 \end{pmatrix}$$

1. По заданным координатам точек A, B, C, D найдем координаты векторов $a=AB$ и $b=CD$.
Занесем координаты данных точек в Excel.

Для этого создадим новый лист и назовем его

«массивы». В ячейку A2 запишем «A», в ячейки B1:B3

заполним значения координат точки A. В ячейку D2

запишем «B», в ячейки E1:E3 заполним значения

координат точки B. Аналогично для точки C заполняем

ячейки: A6 и B5:B7, для точки D – D6, E5:E7.

В ячейку A10 запишем «a», выделим ячейки B9:B11 и в строку формул запишем:

$$=E1:E3-B1:B3$$

после чего нажмем Ctrl+Shift+Enter.

Получили вектор $a=(-3; 5; -6)$.

Для вычисления вектора b в ячейку D10 набираем «b», далее выделяем ячейки E9:E11 и в строке формул и набираем:

$$=E5:E7-B5:B7$$

Получили вектор $b=(-3; 2; 5)$.

2. Вычислим скалярное произведения векторов $a \cdot b$, $b \cdot a$.

Для этого в ячейку A15 наберем «a*b=», а в ячейку B15 наберем формулу:

=МУМНОЖ(ТРАНСП(B9:B11);E9:E11)

нажмем Ctrl+Sift+Enter.

Для получения скалярного произведения $b \cdot a$ в ячейку D15 наберем « $b \cdot a$ », а в ячейку E15:

=МУМНОЖ(ТРАНСП(E9:E11);B9:B11)

нажмем Ctrl+Sift+Enter.

В обоих случаях получилось -11 .

3. Найдем произведения векторов на заданную матрицу M : $a \cdot M$ и $M \cdot b$.

Для этого заполним заданные матрицы. В ячейку G2 запишем « M », а в ячейки H1:J3 заполним значения матрицы. Аналогично заполним значения матрицы S в ячейки G6 и H5:J7.

В ячейку L2 запишем « $a \cdot M$ », выделим ячейки M2:O2 и в строке формул запишем:

=МУМНОЖ(ТРАНСП(B9:B11);H1:J3)

Нажмем Ctrl+Sift+Enter.

В ячейку L6 запишем « $M \cdot b$ », выделим ячейки M5:M7 и в строку формул наберем:

=МУМНОЖ(H1:J3;E9:E11)

Нажмем Ctrl+Sift+Enter.

4. Вычислим определители матриц M и S .

Для этого в ячейку L10 наберем « $|M|$ », а в ячейку M10 формулу:

=МОПРЕД(H1:J3)

и нажмем Ctrl+Sift+Enter.

Затем в ячейку L14 наберем « $|S|$ », а в ячейку M14

формулу:

=МОПРЕД(H5:J7)

и нажмем Ctrl+Sift+Enter.

5. Найдем обратные матрицы S^{-1} и M^{-1} . Определители этих матриц не равны 0.

Для этого в ячейку G10 запишем « $1/M$ », выделим ячейки H9:J11 и в строку формул поместим:

=МОБР(H1:J3)

нажмем Ctrl+Sift+Enter.

В ячейку G14 заполним « $1/S$ », выделим ячейки H13:J15, в строку формул запишем:

=МОБР(H5:J7)

и нажмем Ctrl+Sift+Enter.

Мы получим две обратные матрицы.

6. Для проверки правильности вычисления обратной матрицы вычислим произведение матрицы на обратную к ней.

Для этого в ячейку G18 заполним « $M \cdot 1/M$ », выделим

ячейки H17:J19 и в строке формул наберем:

=МУМНОЖ(H1:J3;H9:J11)

нажмем Ctrl+Sift+Enter.

Аналогично для матрицы S заполним ячейки G22 и

H21:J23. Зададим для матриц числовой формат.

Получили единичные матрицы. Значит, вычисления проведены верно.

7. Найти решение системы линейных уравнений $Mx=a$ и $Sx=b$.

Сначала решим первую систему, для этого в ячейку A21 заполним «x1=», выделим ячейки B20:B22 и в строку

формул запишем:

=МУМНОЖ(H9:J11;B9:B11)

нажмем Ctrl+Sift+Enter.

Решение второй системы получим в ячейки E20:E22

=МУМНОЖ(H13:J15;E9:E11)

8. Выполним проверку для найденных решений.

Для этого вычислим следующие значения: $|M*x1-a|$ и $|S*x2-b|$.

В ячейку A26 заполним « $|M*x1-a|=$ », выделим

ячейки B25:B27 и в строку формул запишем:

=ABS(МУМНОЖ(H1:J3;B20:B22)-B9:B11)

нажмем Ctrl+Sift+Enter.

Аналогично, для проверки решения второй системы

выделим ячейки E25:E27 и в строку формул запишем:

=ABS(МУМНОЖ(H5:J7;E20:E22)-E9:E11)

нажмем Ctrl+Sift+Enter.

Установим этим значениям числовой формат.

В результате получим столбики из нулей.

Следовательно, решение верное.